

```
> restart: with (linalg):
Warning, new definition for norm
Warning, new definition for trace
```

CARTAN'S REPERE MOBILE

A MAPLE PROGRAM to construct the Repere Mobile, or moving Frame Matrix, the Cartan matrix, the Shape Matrix, the Mean Curvature, and Gauss Curvature for a parametric or an implicit 2-surface in R3.

R. M. Kiehn

<http://www.uh.edu/~rkiehn>

This program should be used with the notes given in

<http://www.uh.edu/~rkiehn/pdf/defects2.pdf>

```
> with(diffforms):
> with(liesyymm): setup(u,v,s):
Warning, new definition for `&^`
Warning, new definition for close
Warning, new definition for d
Warning, new definition for mixpar
Warning, new definition for wdegree
```

THE PARAMETRIC METHOD

The Position Vector in R3 parametrized with (u,v). The example is for a Monge Surface $z=g(u,v)$

```
> XX:=u;YY:=v;
XX := u
YY := v
> ZZ:=g(u,v);rho:=scale(u,v);
ZZ := g(u, v)
rho := scale(u, v)
```

Now you can specify formats for $g(u,v)$; $scale(u,v)$ and evaluate specific examples. You could also plot the surfaces. For example just uncheck the line below, or change XX and YY to be specific functions of u and v.

By pass the next step to get an abstract formulation.

```
> ZZ:=u*v^3-v;
ZZ := u v^3 - v
> rho:=1;
```

The scaling is set to 1 here, but see the equation for rho below:

The Torsion terms of the type 2 depend upon the scaling.

$$\rho := 1$$

The position vector in R3

```
> RR:=[XX,YY,ZZ];
RR := [u, v, u v^3 - v]
> Yu:=diff(RR,u);
Yu := [1, 0, v^3]
> Yv:=diff(RR,v);
Yv := [0, 1, 3 u v^2 - 1]
> NNU:=crossprod(Yu,Yv);
```

$$NNU := [-v^3, 1 - 3 u v^2, 1]$$

Scale the adjoint normal field here by rho

> `rho := (NNU[1]^2 + NNU[2]^2 + NNU[3]^2)^(1/2);`

$$\rho := \sqrt{v^6 + (1 - 3 u v^2)^2 + 1}$$

This vector (surface normal) NNU

can be computed from the Adjoint Matrix operation on the two tangent vectors Yu and Yv.

The basis frame utilizes this surface normal with arbitrary scaling.

> `NN := ([factor(NNU[1]), factor(NNU[2]), simplify(factor(NNU[3]))]);`

$$NN := [-v^3, 1 - 3 u v^2, 1]$$

THE IMPLICIT METHOD

Note that the process could have started from the specification of a global Action 1-form: That is define:

> `Action := NNU[1]&d(u) + NNU[2]&d(v) + 1*d(s);`

$$Action := -v^3 d(u) + (1 - 3 u v^2) d(v) + d(s)$$

Now find the associated vectors that annihilate the Action form, and scale the Action 1-form by dividing by rho. Either procedure results in the Basis Frame given below. The Cartan 1-form of Action method will also apply to 1-forms which are not exact (Surfaces which are NOT simple Monge surfaces.)

THE FRAME FF

> `FF := array([[Yu[1], Yv[1], NN[1]/rho], [Yu[2], Yv[2], NN[2]/rho], [Yu[3], Yv[3], NN[3]/rho]]);`

$$FF := \begin{bmatrix} 1 & 0 & -\frac{v^3}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} \\ 0 & 1 & \frac{1 - 3 u v^2}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} \\ v^3 & 3 u v^2 - 1 & \frac{1}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} \end{bmatrix}$$

The Repere Mobile or FRAME MATRIX, FF. note that the frame matrix is not orthonormal!!

> `detFF := simplify((det(FF)));`

$$detFF := \sqrt{v^6 + 2 - 6 u v^2 + 9 u^2 v^4}$$

For the Projective Cartan Monge case, the determinant is non-zero globally, hence an inverse always exists.

> `FFINVD := evalm(FF^(-1));`

$$FFINVD := \begin{bmatrix} \frac{2 - 6 u v^2 + 9 u^2 v^4}{\%1} & -\frac{v^3 (3 u v^2 - 1)}{\%1} & \frac{v^3}{\%1} \\ -\frac{v^3 (3 u v^2 - 1)}{\%1} & \frac{1 + v^6}{\%1} & \frac{3 u v^2 - 1}{\%1} \\ -\frac{v^3}{\sqrt{\%1}} & -\frac{3 u v^2 - 1}{\sqrt{\%1}} & \frac{1}{\sqrt{\%1}} \end{bmatrix}$$

$$\%1 := v^6 + 2 - 6 u v^2 + 9 u^2 v^4$$

The 1-form components of the differential position vector with respect to the Basis Frame, F. Compare the results below to equation 3.20 in the notes.

> `dR:=innerprod(FFINVD,[d(XX),d(YY),d(ZZ)]);`

$$dR := [d(u), d(v), 0]$$

> `sigma1:=wcollect(dR[1]);`

$$\sigma_1 := d(u)$$

> `sigma2:=wcollect(dR[2]);`

$$\sigma_2 := d(v)$$

[Note that sigma1 is du and sigma2 is dv for a parametric Monge surfaces!!

> `omega:=wcollect(dR[3]);`

$$\omega := 0$$

[Note that this term **vanishes** for a parametric Monge surface, hence parametric Monge surfaces exhibit no TORSION!! of the Affine type (that is there is no translational shear defects!) see eq 3.28 of the Notes.

[You should repeat these calculations for a non-Monge surface.

>

[Compute the Cartan Matrix of connection forms from C=[F(inverse)] times d[F]. See equation 3.21 in the notes.

> `dFF:=array([[d(FF[1,1]),d(FF[1,2]),d(FF[1,3])],[d(FF[2,1]),d(FF[2,2]),d(FF[2,3])],[d(FF[3,1]),d(FF[3,2]),d(FF[3,3])]]);`

$$dFF := \begin{bmatrix} 0 & 0 & -3 \frac{v^2 d(v)}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} - v^3 \%1 \\ 0 & 0 & \frac{-3 d(u) v^2 - 6 u v d(v)}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} + (1 - 3 u v^2) \%1 \\ 3 v^2 d(v) & 3 d(u) v^2 + 6 u v d(v) & \%1 \end{bmatrix}$$

$$\%1 := 3 \frac{(1 - 3 u v^2) v^2 d(u)}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} - \frac{1}{2} \frac{(6 v^5 - 12 (1 - 3 u v^2) u v) d(v)}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}}$$

> `cartan:=evalm(FFINVD*dFF);`

`cartan :=`

$$\begin{bmatrix} 3 \frac{v^5 d(v)}{\%1}, \frac{v^3 (3 d(u) v^2 + 6 u v d(v))}{\%1}, \\ \frac{(2 - 6 u v^2 + 9 u^2 v^4) \left(-3 \frac{v^2 d(v)}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} - v^3 \%2 \right)}{\%1} - \frac{v^3 (3 u v^2 - 1) \%3}{\%1} + \frac{v^3 \%2}{\%1} \end{bmatrix}$$

$$\begin{bmatrix} 3 \frac{(3 u v^2 - 1) v^2 d(v)}{\%1}, \frac{(3 u v^2 - 1) (3 d(u) v^2 + 6 u v d(v))}{\%1}, \\ - \frac{v^3 (3 u v^2 - 1) \left(-3 \frac{v^2 d(v)}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} - v^3 \%2 \right)}{\%1} + \frac{(1 + v^6) \%3}{\%1} + \frac{(3 u v^2 - 1) \%2}{\%1} \end{bmatrix}$$

$$\begin{bmatrix} 3 \frac{v^2 d(v)}{\sqrt{\%1}}, \frac{3 d(u) v^2 + 6 u v d(v)}{\sqrt{\%1}}, - \frac{v^3 \left(-3 \frac{v^2 d(v)}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} - v^3 \%2 \right)}{\sqrt{\%1}} - \frac{(3 u v^2 - 1) \%3}{\sqrt{\%1}} + \frac{\%2}{\sqrt{\%1}} \end{bmatrix}$$

$$\%1 := v^6 + 2 - 6 u v^2 + 9 u^2 v^4$$

$$\%2 := 3 \frac{(1 - 3 u v^2) v^2 d(u)}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} - \frac{1}{2} \frac{(6 v^5 - 12 (1 - 3 u v^2) u v) d(v)}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}}$$

$$\%3 := \frac{-3 d(u) v^2 - 6 u v d(v)}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} + (1 - 3 u v^2) \%2$$

The interior connection coefficients (can be Christoffel symbols on the parameter space

> **Gamma11:=wcollect(cartan[1,1]);**

$$\Gamma_{11} := 3 \frac{v^5 d(v)}{v^6 + 2 - 6 u v^2 + 9 u^2 v^4}$$

> **Gamma12:=wcollect(cartan[1,2]);**

$$\Gamma_{12} := 3 \frac{v^5 d(u)}{v^6 + 2 - 6 u v^2 + 9 u^2 v^4} + 6 \frac{v^4 u d(v)}{v^6 + 2 - 6 u v^2 + 9 u^2 v^4}$$

> **Gamma21:=wcollect(cartan[2,1]);**

$$\Gamma_{21} := 3 \frac{(3 u v^2 - 1) v^2 d(v)}{v^6 + 2 - 6 u v^2 + 9 u^2 v^4}$$

> **Gamma22:=wcollect(cartan[2,2]);**

$$\Gamma_{22} := 3 \frac{(3 u v^2 - 1) v^2 d(u)}{v^6 + 2 - 6 u v^2 + 9 u^2 v^4} + 6 \frac{(3 u v^2 - 1) u v d(v)}{v^6 + 2 - 6 u v^2 + 9 u^2 v^4}$$

The second fundamental form comes from the h row vector and the shape matrix comes from the gamma column vector of the Cartan matrix

> **h1:=wcollect(cartan[3,1]);**

$$h_1 := 3 \frac{v^2 d(v)}{\sqrt{v^6 + 2 - 6 u v^2 + 9 u^2 v^4}}$$

> **gamma1:=wcollect(cartan[1,3]);**

$$\gamma_1 := \left(-3 \frac{(2 - 6 u v^2 + 9 u^2 v^4) v^5 (1 - 3 u v^2)}{\%1 (v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right. \\ \left. - \frac{v^3 (3 u v^2 - 1) \left(-3 \frac{v^2}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} + 3 \frac{(1 - 3 u v^2)^2 v^2}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right)}{\%1} + 3 \frac{v^5 (1 - 3 u v^2)}{\%1 (v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right) \\ d(u) + \left(\frac{(2 - 6 u v^2 + 9 u^2 v^4) \left(-3 \frac{v^2}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} + \frac{1}{2} \frac{v^3 (6 v^5 - 12 (1 - 3 u v^2) u v)}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right)}{\%1} \right. \\ \left. - \frac{v^3 (3 u v^2 - 1) \left(-6 \frac{u v}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} - \frac{1}{2} \frac{(1 - 3 u v^2) (6 v^5 - 12 (1 - 3 u v^2) u v)}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right)}{\%1} \right) \\ \left. - \frac{1}{2} \frac{v^3 (6 v^5 - 12 (1 - 3 u v^2) u v)}{\%1 (v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right) d(v) \\ \%1 := v^6 + 2 - 6 u v^2 + 9 u^2 v^4$$

```
> h2:=(wcollect(cartan[3,2]));
```

$$h2 := 3 \frac{v^2 d(u)}{\sqrt{v^6 + 2 - 6 u v^2 + 9 u^2 v^4}} + 6 \frac{u v d(v)}{\sqrt{v^6 + 2 - 6 u v^2 + 9 u^2 v^4}}$$

```
> gamma2:=(wcollect(cartan[2,3]));
```

$$\gamma_2 := \left(3 \frac{v^8 (3 u v^2 - 1) (1 - 3 u v^2)}{\%1 (v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} + \frac{(1 + v^6) \left(-3 \frac{v^2}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} + 3 \frac{(1 - 3 u v^2)^2 v^2}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right)}{\%1} \right) \\ + 3 \frac{(3 u v^2 - 1) (1 - 3 u v^2) v^2}{\%1 (v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} d(u) + \left(\frac{v^3 (3 u v^2 - 1) \left(-3 \frac{v^2}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} + \frac{1}{2} \frac{v^3 (6 v^5 - 12 (1 - 3 u v^2) u v)}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right)}{\%1} \right) \\ - \frac{(1 + v^6) \left(-6 \frac{u v}{\sqrt{v^6 + (1 - 3 u v^2)^2 + 1}} - \frac{1}{2} \frac{(1 - 3 u v^2) (6 v^5 - 12 (1 - 3 u v^2) u v)}{(v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} \right)}{\%1} \\ - \frac{1}{2} \frac{(3 u v^2 - 1) (6 v^5 - 12 (1 - 3 u v^2) u v)}{\%1 (v^6 + (1 - 3 u v^2)^2 + 1)^{3/2}} d(v) \\ \%1 := v^6 + 2 - 6 u v^2 + 9 u^2 v^4$$

The abnormality (Big Omega) for the parametric surface will show up as a non-zero entry in the [3,3] slot of the Cartan Matrix. It is always an exact differential for parametric and Monge surfaces, but vanishes identically if the scaling factor is chosen properly. Therefore Monge or implicit surfaces will admit disclination defects from Omega (Torsion of the second kind due to rotations or expansions) These are not affine torsion coefficients which are due to shears of translation. (Is there Torsion in the Expanding Universe?)

```
> Omega:=(wcollect((factor(cartan[3,3]))));
```

$$\Omega := 0$$

Omega vanishes identically for a Monge surface **if the scaling factor is such as to normalize the Normal Field to unity everywhere**. Note that the determinant of the Frame matrix is not unimodular when Omega vanishes.

If the Omega term vanishes there is no torsion 2-forms of the second type.

How does nature choose the scaling??

See equation 3.30 of the notes

When Omega is not zero, you can compute the terms of the Abnormality Torsion 2-forms from the formulas: (just remove the # symbol)

```
> # wcollect(factor(simpform(d(Omega))));
```

```
> # FROBOMEGA:=simpform(Omega&^d(Omega));
```

The coefficients of the torsion 2-forms of the second type.

```
> # factor(simpform(Omega&^gamma1));
```

```
[ > # simplify(Omega^gamma2);
```

[>

[To compute the elements of the shape matrix (with trace = mean curvature, and determinant = Gauss curvature)

```
[ > shape11:=-factor(gamma1&^d(v)/d(u)&^d(v));
```

$$shape11 := -3 \frac{v^5 (3 u v^2 - 1)}{(v^6 + 2 - 6 u v^2 + 9 u^2 v^4)^{3/2}}$$

```
[ > shape12:=-factor(gamma1&^d(u)/d(v)&^d(u));
```

$$shape12 := 3 \frac{v^2 (2 - 4 u v^2 + 3 u^2 v^4)}{(v^6 + 2 - 6 u v^2 + 9 u^2 v^4)^{3/2}}$$

```
[ > shape21:=-factor(gamma2&^d(v)/d(u)&^d(v));
```

$$shape21 := 3 \frac{v^2 (v^2 + 1) (v^4 - v^2 + 1)}{(v^6 + 2 - 6 u v^2 + 9 u^2 v^4)^{3/2}}$$

```
[ > shape22:=-factor(gamma2&^d(u)/d(v)&^d(u));
```

$$shape22 := -3 \frac{v (-2 u - v^4 + u v^6)}{(v^6 + 2 - 6 u v^2 + 9 u^2 v^4)^{3/2}}$$

```
[ >
```

```
[ > SHAPE:=array([[shape11,shape12],[shape21,shape22]]):
```

```
[ > HH:=simplify(trace(SHAPE)/2):
```

```
[ > print(`Mean Curvature is `,HH);
```

$$\text{Mean Curvature is , } -3 \frac{v (2 u v^6 - v^4 - u)}{(v^6 + 2 - 6 u v^2 + 9 u^2 v^4)^{3/2}}$$

```
[ > KK:=simplify(det(SHAPE));
```

$$KK := -9 \frac{v^4}{(v^6 + 2 - 6 u v^2 + 9 u^2 v^4)^2}$$

```
[ > print(`Gauss Curvature is `,KK);
```

$$\text{Gauss Curvature is , } -9 \frac{v^4}{(v^6 + 2 - 6 u v^2 + 9 u^2 v^4)^2}$$

[Note that the scaling of the normal or adjoint vector is a common factor of the formulas for the mean curvature and the Gauss curvature. Note the appearance of the Hessian of the Monge function.

[The induced metric becomes:

```
[ > GUN:=innerprod(transpose(FF),FF);
```

$$GUN := \begin{bmatrix} 1 + v^6 & v^3 (3 u v^2 - 1) & 0 \\ v^3 (3 u v^2 - 1) & 2 - 6 u v^2 + 9 u^2 v^4 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$


[The induced metric;

[You could plot the surfaces here if you wanted too.

[For example, a plot of the Gauss Curvature over the plane of surface variables:

```
[ > plot3d(-KK,u=-0.5..2,v=-1.5..1.5,numpoints=5000,title=`Gauss Curvature`,axes=FRAMED,shading=ZGREYSSCALE,style=PATCHCONTOUR);
```

Gauss Curvature



- >
- >
- >
- >